*===============================================================================
* Title: Two Levels - cross section final vote regional election
* Date: Jan 23 2017
* Author: Agatha Kratz
*===============================================================================

*** Abbildung 1: Wählerwanderung BTW LTW
tab wabsfull8 Niwahl9


*** globals für die Regression
global socdem 					"female age i.edu"
global B_candidates9			"B_candcdu9 B_candspd9 B_candfdp9 B_candgru9 B_candlin9 B_candafd9"
global L_candidates9			"L_candcdu9 L_candspd9 L_candfdp9 L_candgru9 L_candlin9 L_candafd9"
global candidates9				"$B_candidates9	$L_candidates9"
global Lparty 					"cdu spd fdp gru lin afd"
global pid						"keine" "cdu spd fdp gru lin afd"


* Stärke der Parteiidentifikation (wenn fehlend in Welle 9, dann aus Welle 8)
foreach party in union spd fdp gru lin afd {
clonevar	`party'pids9_imp	=	`party'pids9
replace		`party'pids9_imp	=	`party'pids8 if missing(`party'pids9_imp)
}
clonevar cdupids9_imp = unionpids9_imp // Variablennamen anpassen, um Schleife zu ermöglichen

* AfD-Bundespolitiker: Gauland statt Petry
drop B_candafd9
gen B_candafd9 = (kp9_650j-1)/10
replace B_candafd9 	=	.p	if kp9_650j	==	.p


*** Variablenlabels für die Regression
lab var age 			"Alter"
lab var female 			"Frau"
lab def edu 0 			"Bildung: niedrig" 1 "Bildung: mittel" 2 "Bildung: hoch", modify

lab var B_candcdu9 	"Angela Merkel"
lab var B_candspd9 	"Martin Schulz"
lab var B_candfdp9 	"Christian Lindner"
lab var B_candgru9 	"Katrin Göring-Eckardt"
lab var B_candlin9 	"Sahra Wagenknecht"
lab var B_candafd9 	"Alexander Gauland"

lab var L_candcdu9 	"Bernd Althusmann"
lab var L_candspd9 	"Stephan Weil"
lab var L_candfdp9 	"Stefan Birkner"
lab var L_candgru9 	"Anja Piel"
lab var L_candlin9 	"Anja Stoeck"
lab var L_candafd9 	"Dana Guth"




*** Anteile der Befragten, die die Kandidaten nicht kannten (Nachwahlwelle)
fre L_candcdu9 if kp9_650e1==.p | L_candcdu9!=.
sum L_candcdu9
fre L_candspd9 if kp9_650d1==.p | L_candspd9!=.
sum L_candspd9
fre L_candfdp9 if kp9_650f1==.p | L_candfdp9!=.
sum L_candfdp9
fre L_candgru9 if kp9_650g1==.p | L_candgru9!=.
sum L_candgru9
fre L_candlin9 if kp9_650h1==.p | L_candlin9!=.
sum L_candlin9
fre L_candafd9 if kp9_650k1==.p | L_candafd9!=.
sum L_candafd9

fre  L_dkcand5 L_dkcand6 L_dkcand7 L_dkcand8 L_dkcand9
/* In der letzten Befragung vor der LTW konnten 62 Prozent mindestens 
drei der fünf Kandidaten bewerten, 53 Prozent konnten alle fünf bewerten. 

In die Querschnittsregression gehen 62 Prozent der niedersächsischen Befragten
mit einer Teilnahme in der Nachwahlwelle ein, die alle Politiker bewerten konnten.
*/





*** Logistische Regressionen für die beiden Regierungs- und Oppositionsparteien
*** ohne Nichtwähler
est clear
foreach party in $Lparty {
logit L_vote`party'9 `party'pids9_imp $candidates9 $socdem if NIwahl9!=9999, vce(robust)
margins, dydx(*) post
gen obs_`party'=1 if e(sample)==1
estimates store logit_`party'
}
esttab	logit_cdu logit_spd logit_fdp logit_gru logit_lin logit_afd using "tables\Tabelle 3_Querschnittsregression_test.rtf", replace ///
		title("Tabelle X: Einfluss landes- und bundespolitischer Faktoren auf die Wahlentscheidung in der niedersächsischen Landtagswahl (logistische Regressionen)") ///
		mtitles("CDU" "SPD" "FDP" "Grüne" "Die Linke" "AfD") b(2) se pr2(2) nolz nogaps nonotes obslast drop(0.edu) star(* 0.05 ** 0.01) ///
		addnotes("Anmerkung: Dargestellt sind die durchschnittlichen marginale Effekte mit robusten Standardfehler in Klammern. * Signifikant auf dem 95%-Niveau, ** signifikant auf dem 99%-Niveau.") /// 
		varwidth(30) modelwidth(3) label nonotes align(l)

*** Varianzaufklärung
* Gesamtmodell
foreach party in $Lparty {
logit L_vote`party'9 `party'pids9_imp $candidates9 $socdem if NIwahl9!=9999, vce(robust)
fitstat
}
		
* Modell nur mit PID
foreach party in $Lparty {
logit L_vote`party'9 `party'pids9_imp $socdem if NIwahl9!=9999 & obs_`party'==1, vce(robust)
fitstat
}

* Modell nur mit den Bundeskandidaten
foreach party in $Lparty {
logit L_vote`party'9 $B_candidates9 $socdem if NIwahl9!=9999 & obs_`party'==1, vce(robust)
fitstat
}
		
* Modell nur mit den Landeskandidaten
foreach party in $Lparty {
logit L_vote`party'9 $L_candidates9 $socdem if NIwahl9!=9999 & obs_`party'==1, vce(robust)
fitstat
}	
		
* Modell nur mit den Bundeskandidaten + PID
foreach party in $Lparty {
logit L_vote`party'9 `party'pids9_imp $B_candidates9 $socdem if NIwahl9!=9999 & obs_`party'==1, vce(robust)
fitstat
}	
	
* Modell nur mit den Landeskandidaten + PID
foreach party in $Lparty {
logit L_vote`party'9 `party'pids9_imp $L_candidates9 $socdem if NIwahl9!=9999 & obs_`party'==1, vce(robust)
fitstat
}	
	
drop cdupids9	


* AfD-Bundespolitiker: Gauland wieder durch Petry ersetzen
drop B_candafd9
gen B_candafd9 = (kp9_650x-1)/10
replace B_candafd9 	=	.p	if kp9_650x	==	.p
		
		
		
		
		